/*
 * Flashlib - QB9 common Flash library
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

package com.qb9.flashlib.security
{
	import flash.utils.ByteArray;
	
	/**
	 * Stores a number safely, making it hard to find in memory (using CheatEngine).
	 * The values can be integers with or without sign, float numbers and even NaN.
	 * SafeNumbers can be used directly as numbers.
	 */
	
	public final class SafeNumber
	{
		private const bytes:ByteArray = new ByteArray();
		
		public function SafeNumber(value:Number=NaN)
		{
			this.value = value;
		}
		
		private function reset():void
		{
			bytes.position = 0;
		}
		
		public function get value():Number
		{
			if (bytes.length === 0)
				return NaN;
			
			reset();
			return bytes.readFloat();
		}
		
		public function set value(v:Number):void
		{
			reset();
			if (isNaN(v))
				bytes.length = 0;
			else
				bytes.writeFloat(v);
		}
		
		public function valueOf():Object
		{
			return value;
		}

	}
}